
Shortest-path grocery shopping
What’s the shortest path thru the grocery store to get my items?

Justin Pearson
Apr 19, 2020

������� Clear["Global`*"]
SetDirectory[NotebookDirectory[]];

Summary
Given an image of the store’s aisles, we mesh it and generate a graph from the mesh. Then we use 

various shortest-path algorithms on the graph to find the shortest tour that visits all the items.

We are careful to build a graph whose edge-weights equal the euclidean distance between adjacent 
vertices, so that that shortest-path is shortest with respect to actual space, not just “number of hops on 

the graph”.

������� i = ;



������� m = ImageMesh[i]

�������

Triangularize the mesh. Pick a small “max cell size” to get small triangles.

������� Manipulate[
TriangulateMesh[m, MaxCellMeasure → c],
{{c, 100, "Max cell size (px)"}, 10, 500, 1, Appearance → "Open"}

]

�������

��� ���� ���� (��)

���

������� t = TriangulateMesh[m, MaxCellMeasure → 100]

�������

Convert the triangularized-mesh to a graph, using the degree-0 elements of the mesh -- the mesh 

points -- as vertices. (Degree-1 is the triangle edges, degree-2 is the triangles themselves.)

������� g = MeshConnectivityGraph[t, 0]

�������

A slick UI for showing shortest-paths between any two vertices.

������� Manipulate[
GraphPlot[
HighlightGraph[
g,
PathGraph@
FindShortestPath[
g,
NearestMeshCells[{t, 0}, u〚1〛, 1] // First,
NearestMeshCells[{t, 0}, u〚2〛, 1] // First

],
GraphHighlightStyle → "DehighlightGray"],

EdgeStyle → Thickness[0.005],
ImageSize → 400

],
{{u, {{100, 100}, {200, 200}}}, Locator}

]

�������

We extend this idea to build a shortest-path thru the store that visits all your items:

������� Show[First@Import["map-1200px.pdf"], ImageSize → Full]

�������

- This is a real map of my local Albertsons grocery store.
- The thin pink line is the shortest tour, ignoring occlusions like the (black) aisles.
- The colored paths visit the items in the order of the shortest tour. 
- To find the shortest tour thru the items along the triangulated mesh, I first found the shortest pairwise 

distance between every pair of items on the triangulated-mesh graph. Then I made a new complete 

graph with only 40 vertices -- the items -- whose edge weights were the pairwise distances. Then I 
asked MMA to find the shortest tour of those 40 vertices. (It would have been simpler to ask MMA to find 

the shortest path on the triangulated-mesh graph from vertex 1 to vertex 40 that visits all 40 vertices, 
but I didn’t know how to do that.)

Build graph of the grocery store
�������� albertsons = Import["albertsons-map/albertsons-map.001.jpeg"]

��������

�������� im = albertsons // Binarize // ColorNegate //

DeleteSmallComponents[#, 700] & // ColorNegate;
Thumbnail[
im,
700]

��������

�������� mesh = ImageMesh[im]

��������

�������� tmesh = TriangulateMesh[mesh, MaxCellMeasure → 100];
Show[tmesh, ImageSize → Full]

��������

�������� meshGraph = MeshConnectivityGraph[tmesh, 0]

�������� Graph ������ ������ �����
���� ������ �����



�������� GraphPlot[meshGraph, ImageSize → Full]

��������

Get the graph vertices. 
Notice that the vertices are not integers, but rather of the form {0,3}. Read this as “Point number 3”. 
(Because vertices correspond to points on the mesh, not lines or faces, and points are “degree 0”.)

�������� vlist = VertexList[meshGraph];
Length@vlist
vlist // Short

�������� 14150

��������������� {{0, 1}, {0, 2}, {0, 3}, {0, 4}, {0, 5},
14141, {0, 14147}, {0, 14148}, {0, 14149}, {0, 14150}}

�������� elist = EdgeList[meshGraph];
Length@elist
elist // Short

�������� 40299

��������������� {{0, 1}  {0, 13897}, {0, 1}  {0, 13899},
40295, {0, 14145}  {0, 14146}, {0, 14145}  {0, 14147}}

Here are the euclidean coordinates of each vertex.

�������� vcoords = GraphEmbedding[meshGraph];
vcoords // Short

��������������� {{1920., 1080.}, {0., 1080.}, 14147, {784.269, 266.766}}

shoptimization-v5_WEB.nb  ���9



Map vertices to their coordinates.

�������� v2coord = Association @@ MapThread[Rule, {vlist, vcoords}]

��������

{0, 1} → {1920., 1080.}, {0, 2} → {0., 1080.},

{0, 3} → {0., 0.}, {0, 4} → {1920., 0.}, ⋯ 14142⋯ ,
{0, 14147} → {745.668, 305.828}, {0, 14148} → {770.708, 307.776},

{0, 14149} → {748.463, 280.563}, {0, 14150} → {784.269, 266.766}

����� ������ ���� ���� ���� ���� ���� ��� ��� ���� ��������

Build a new graph with the same vertices and edges, but whose edges are weighted by the euclidean 

distance between the vertices.

�������� AbsoluteTiming[
weightedMeshGraph = Graph[

vlist,
elist,
EdgeWeight → {u_  v_ ⧴ EuclideanDistance[v2coord[u], v2coord[v]]},
VertexCoordinates → vcoords

];
]

�������� {90.4457, Null}

The weighted graph has edges that are weighted by the euclidean distance between the vertices.
This is important for distance-finding. Here are two adjacent vertices:

�������� {v1, v2} = List @@ First@elist

�������� {{0, 1}, {0, 13897}}

In the raw mesh graph, they are only a distance “1” apart:

�������� GraphDistance[meshGraph, v1, v2]

�������� 1

In the weighted graph, their graph distance is the euclidean distance:

�������� GraphDistance[weightedMeshGraph, v1, v2]

�������� 16.875

�������� v2coord /@ {v1, v2}

�������� {{1920., 1080.}, {1920., 1063.13}}

�������� EuclideanDistance @@ %

�������� 16.875

Shopping trip
�������� header = Import["Grocery list - items.csv"] // First

�������� {Name, Category, Location in Kitchen, aisle, front/mid/back/backwall/endcap,
left/right, hi/med/lo, Notes, x-coordinate, y-coordinate, 4/18/2020 - need it?}

�������� rows = Import["Grocery list - items.csv"] // Cases[{___, _Integer, _Integer, ___}];

�������� numAllItems = Length@rows
Short[rows, 10]

�������� 89

��������������� {{entrance, n/a, n/a, 1, F, , , , 1399, 96, },
{red wine, booze, over-toaster-cupboard, 2, M, R, M, , 1549, 755, y},
{greek yogurt, dairy, fridge, 4, BW, , , whole milk, large, 1381, 984, y},
{paper towels, paper products, coat closet, 5, B, L, , , 1286, 686, y},
{cadburry chocolate eggs, candy, pantry, 6, F, L, , , 1212, 291, y},
80, {tomatoes, produce, fridge, produce, , , , , 343, 130, y},
{yellow onion, produce, counter, produce, , , , , 336, 172, y},
{apples, produce, fridge, produce, , , , , 448, 124, y},
{avocado, produce, counter, produce, , , , , 379, 124, y}}

�������� entranceCoord =

rows // FirstCase[{"entrance", ___, x_Integer, y_Integer, ___} ⧴ {x, y}];
exitCoord = rows // FirstCase[{"exit", ___, x_Integer, y_Integer, ___} ⧴ {x, y}];

�������� {iName, iXcoord, iYcoord, iNeedIt} = Table

First@FirstPosition[header, patt, Heads → False], patt,

"Name", "x-coordinate", "y-coordinate", _?StringContainsQ["need it?"]

;

allItemNames = rows〚All, iName〛;
allItemCoords = N@rows〚All, {iXcoord, iYcoord}〛;
shoppingList = Pick[allItemNames, rows〚All, iNeedIt〛, "y"];
PrependTo[shoppingList, "entrance"];
AppendTo[shoppingList, "exit"];

�������� item2coord = Association@MapThread[Rule, {allItemNames, allItemCoords}];

�������� item2coord["avocado"]

�������� {379., 124.}

�������� item2coord["milk"]

�������� {1146., 985.}

�������� item2coord["entrance"]

�������� {1399., 96.}

�������� gItemNames = Graphics@Table[
Text[
s,
item2coord[s],
Background → If[

MemberQ[shoppingList, s],
Pink,
LightBlue

]

], {s, allItemNames}];

�������� Show[im, gItemNames, ImageSize → Full]

��������

entrance

red wine

greek yogurt

paper towels

cadburry chocolate eggscadburry creme eggs

milkalmond milkwhole milkcoffee creamerheavy cream

eggs

toilet paper

bacon
breakfast sausage
canadian bacon (x3)

butter

parmesean cheese
ricotta cheese
shredded cheese
cheese stick

cream cheese
cream-filled mozzarella

ice cream

froz tater tot

cottage cheese

froz corn
froz peas

beef chuck roastchickenpork buttground turkey

Pirates booty

almonds
peanuts

pork shoulderbeef short ribs

froz shrimp

crackers

fizzy water

pom juice

olive oil

small flour tortillas

flour

maple syrup
whole wheat flour

yeast (active)
corn starch

rice

pasta sauce

pasta - bows
pasta - egg noodles

canned salmon

canned tuna

coconut milk - can

hot chocolate mix

mustard

peanut butter

bagels

bread

exit

Poppyseed salad dressing

hummusfancy cheese

feta cheese

sushi - CA roll

prosciutto

bag kale

blueberries

carrots

cauliflower
cucumber

fresh rosemarygarlicdiced jar garlic

grapes

lemon juicelimes
orange

potatoes
red onionshallot

snap peas

strawberries
sweet potatoes

tomatoes

yellow onion

applesavocado

�������� itemCoords = item2coord /@ shoppingList

�������� {{1399., 96.}, {1549., 755.}, {1381., 984.}, {1286., 686.}, {1212., 291.},
{1212., 306.}, {1146., 985.}, {1113., 985.}, {1127., 985.}, {1067., 987.},
{1054., 988.}, {1033., 897.}, {1010., 813.}, {1012., 794.}, {988., 985.},
{1011., 403.}, {1011., 443.}, {1011., 472.}, {1011., 540.}, {901., 702.},
{915., 988.}, {767., 991.}, {841., 988.}, {801., 986.}, {826., 989.},
{855., 726.}, {746., 776.}, {747., 580.}, {779., 655.}, {780., 746.},
{649., 891.}, {669., 527.}, {668., 485.}, {592., 306.}, {590., 420.},
{550., 279.}, {473., 554.}, {394., 289.}, {394., 335.}, {360., 743.},
{1631., 277.}, {1571., 278.}, {1632., 312.}, {388., 91.}, {457., 168.},
{342., 91.}, {287., 137.}, {288., 103.}, {286., 167.}, {368., 178.}, {421., 200.},
{385., 141.}, {502., 173.}, {376., 204.}, {342., 187.}, {354., 175.}, {461., 186.},
{358., 214.}, {343., 130.}, {336., 172.}, {448., 124.}, {379., 124.}, {631., 94.}}

�������� itemVertices = First@NearestMeshCells[{tmesh, 0}, #, 1] & /@ itemCoords

�������� {{0, 4861}, {0, 6989}, {0, 9420}, {0, 8101}, {0, 7142}, {0, 7147}, {0, 8868},
{0, 7834}, {0, 7831}, {0, 6206}, {0, 6200}, {0, 7222}, {0, 10152}, {0, 10140},
{0, 6131}, {0, 7115}, {0, 7134}, {0, 7500}, {0, 9873}, {0, 7742}, {0, 6138},
{0, 6169}, {0, 5996}, {0, 6162}, {0, 6156}, {0, 5874}, {0, 11070}, {0, 14004},
{0, 13980}, {0, 11086}, {0, 6279}, {0, 4890}, {0, 4884}, {0, 7408}, {0, 11568},
{0, 6510}, {0, 5628}, {0, 7047}, {0, 194}, {0, 8043}, {0, 5115}, {0, 5139},
{0, 5118}, {0, 344}, {0, 429}, {0, 364}, {0, 5709}, {0, 5723}, {0, 373},
{0, 184}, {0, 3215}, {0, 304}, {0, 452}, {0, 173}, {0, 204}, {0, 272},
{0, 432}, {0, 205}, {0, 148}, {0, 188}, {0, 418}, {0, 335}, {0, 3747}}

Example

Find a path from the first item (“entrance”)to the last item (“exit”):

�������� p = FindShortestPath[
weightedMeshGraph,
itemVertices〚1〛,
itemVertices〚-1〛

]

�������� {{0, 4861}, {0, 4751}, {0, 4758}, {0, 4768}, {0, 4757}, {0, 4399}, {0, 4398}, {0, 3790},
{0, 4731}, {0, 4219}, {0, 4221}, {0, 4226}, {0, 4191}, {0, 4204}, {0, 4203}, {0, 4122},
{0, 3803}, {0, 4069}, {0, 3892}, {0, 4051}, {0, 3711}, {0, 3878}, {0, 3792}, {0, 3891},
{0, 3881}, {0, 96}, {0, 3567}, {0, 3699}, {0, 3714}, {0, 3481}, {0, 5743}, {0, 3719},
{0, 5744}, {0, 3488}, {0, 3651}, {0, 3342}, {0, 3648}, {0, 3772}, {0, 3421},
{0, 3771}, {0, 3359}, {0, 3750}, {0, 3755}, {0, 3636}, {0, 3748}, {0, 3747}}

�������� txt = Graphics@Table[
Text[
Style[shoppingList〚i〛, 16, White, Background → Black],
itemCoords〚i〛

], {i, {1, -1}}];

�������� Show[GraphPlot@HighlightGraph[
weightedMeshGraph,
PathGraph[p]

], txt, ImageSize → Full]

��������

entranceexit

Find best order to visit the items
Idea: make a NEW graph, (a complete graph): vertices are the items, edges are weighted by shortest-
dist from the mesh graph.
Then find a tour of this new graph, and map it back to the mesh graph.

�������� AbsoluteTiming[
distMatrix = Table[GraphDistance[weightedMeshGraph, s, t],

{s, itemVertices}, {t, itemVertices}];
]

�������� {34.4538, Null}

�������� itemGraph = WeightedAdjacencyGraph[distMatrix];

Here’s the shortest path thru all the items:

�������� {len, path} = FindShortestTour[itemGraph, 1, Length@shoppingList]

�������� {8567.37, {1, 42, 41, 43, 2, 3, 4, 6, 5, 16, 17, 18, 19, 14, 13, 12, 7, 9, 8, 10, 11, 15, 21,
23, 25, 24, 22, 20, 26, 30, 29, 28, 27, 31, 32, 33, 35, 34, 36, 37, 40, 39, 38,
58, 54, 51, 50, 56, 55, 60, 49, 47, 48, 46, 59, 52, 62, 44, 61, 45, 57, 53, 63}}

�������� shoppingList〚path〛

�������� {entrance, fancy cheese, hummus, prosciutto, red wine, greek yogurt, paper towels,
cadburry creme eggs, cadburry chocolate eggs, parmesean cheese, shredded cheese,
cheese stick, cream cheese, canadian bacon (x3), breakfast sausage,
eggs, milk, whole milk, almond milk, coffee creamer, heavy cream, butter,
cottage cheese, chicken, ground turkey, pork butt, beef chuck roast,
froz tater tot, Pirates booty, pom juice, fizzy water, crackers, peanuts,
small flour tortillas, whole wheat flour, yeast (active), pasta - bows,
pasta sauce, canned salmon, hot chocolate mix, Poppyseed salad dressing,
bread, bagels, sweet potatoes, potatoes, grapes, garlic, shallot, red onion,
yellow onion, fresh rosemary, cauliflower, cucumber, carrots, tomatoes,
limes, avocado, bag kale, apples, blueberries, strawberries, orange, exit}

This should be the same distance as traversing associated vertices on the weighted mesh graph:

�������� GraphDistance[itemGraph, path〚1〛, path〚2〛]

�������� 262.221

�������� GraphDistance[weightedMeshGraph, itemVertices〚path〚1〛〛, itemVertices〚path〚2〛〛]

�������� 262.221

�������� GraphDistance[itemGraph, #1, #2] & @@@

Partition[path, 2, 1] // Total

�������� 8567.37

�������� GraphDistance[weightedMeshGraph, itemVertices〚#1〛, itemVertices〚#2〛] & @@@

Partition[path, 2, 1] // Total

�������� 8567.37

Here’s the shortest path thru the items, if you get to move through the walls:

�������� {lenIgnoringBoundaries, pathIgnoringBoundaries} =

FindShortestTour[itemCoords, 1, Length@shoppingList]

�������� {5818.61, {1, 41, 43, 42, 5, 6, 16, 17, 18, 19, 4, 2, 3, 7, 9, 8, 10, 11, 12, 14, 13, 15, 21,
23, 25, 24, 22, 31, 27, 30, 26, 20, 29, 28, 32, 33, 40, 37, 35, 34, 36, 39, 38,
51, 54, 58, 50, 56, 55, 60, 49, 47, 48, 46, 59, 52, 62, 44, 61, 45, 57, 53, 63}}

However, if you account for the boundaries, then visiting the items in that order is longer than the 

shortest path:

�������� GraphDistance[weightedMeshGraph, itemVertices〚#1〛, itemVertices〚#2〛] & @@@

Partition[pathIgnoringBoundaries, 2, 1] // Total

�������� 9807.88

(The optimal one is less:)

�������� len

�������� 8567.37

Show the optimal path thru the items:

�������� pathGraphs = FindShortestPath[weightedMeshGraph, #1, #2] & @@@

Partition[itemVertices〚path〛, 2, 1] // Map[PathGraph];

�������� highlight2 = GraphPlot[
HighlightGraph[weightedMeshGraph,
pathGraphs, GraphHighlightStyle → "DehighlightHide"],

EdgeStyle → Thickness[0.005], ImageSize → Full];

�������� gItems = Graphics@Table[
With[{pt = itemCoords〚path〛〚i〛}, {

Text[
Style[shoppingList〚path〛〚i〛, Black, 14, Plain, Background → LightBlue], pt]

}], {i, Length@path}];

�������� map = Show[im,
highlight2,
Graphics@{Pink, Line[itemCoords〚pathIgnoringBoundaries〛]},
gItems,
ImageSize → Full,
ImagePadding → 10]

��������

entrance

fancy cheesehummus
prosciutto

red wine

greek yogurt

paper towels

cadburry creme eggscadburry chocolate eggs

parmesean cheese
shredded cheese
cheese stick

cream cheese

canadian bacon (x3)breakfast sausage

eggs

milkwhole milkalmond milkcoffee creamerheavy creambuttercottage cheesechickenground turkeypork buttbeef chuck roast

froz tater totPirates bootypom juice

fizzy water

crackers

peanuts

small flour tortillas

whole wheat flour
yeast (active)

pasta - bows

pasta sauce
canned salmon

hot chocolate mix

Poppyseed salad dressing

bread
bagels

sweet potatoespotatoesgrapesgarlicshallotred onionyellow onionfresh rosemary
cauliflower
cucumbercarrots
tomatoeslimesavocado
bag kale

apples
blueberriesstrawberriesorange

exit

�������� Show[map, PlotRange → {{240, 630}, {50, 280}}]

��������

canned salmon
bagels

sweet potatoes
potatoes grapes

garlicshallot
red onion

yellow onion
fresh rosemary

cauliflower

cucumber
carrots

tomatoes
limes

avocado

bag kale

apples

blueberries

strawberries

orange

exit

�������� Export["map-1200px.pdf", map, ImageSize → 1200]
Speak["Done"]

�������� map-1200px.pdf

